

Temperatursensor og LCD-skjerm

LØSNINGSFORSL
AG

8.-10. trinn · 45 min · SkapLab – Arduino

Hva du skal lære

Lese temperatur fra en TMP36-sensor, konvertere spenning til Celsius og vise på en I2C LCD-skjerm.

Utstyr du trenger

- 1× TMP36 temperatursensor (eller LM35)
- 1× I2C LCD-skjerm 16×2 (adresse 0x27 eller 0x3F)
- Ledninger og breadboard

Koble til slik

PIN	KOMPONENT	BESKRIVELSE
A0	TMP36 – midtbein (Vout)	Analog spenningsutgang fra sensor
5V	TMP36 – venstre bein (+Vs)	Spenningsforsyning til sensor
GND	TMP36 – høyre bein (GND)	Jord for sensor
A4	LCD SDA	I2C data (Uno: A4 = SDA)
A5	LCD SCL	I2C klokke (Uno: A5 = SCL)
5V	LCD VCC	Spennning til LCD
GND	LCD GND	Jord for LCD

Kodeeksempel

```
#include <Wire.h>
#include <LiquidCrystal_I2C.h>

LiquidCrystal_I2C lcd(0x27, 16, 2);

void setup() {
  Serial.begin(9600);
  lcd.init();
  lcd.backlight();
  lcd.setCursor(0, 0);
  lcd.print("Temperatur:");
}
```

```

void loop() {
  int raaverdi = analogRead(A0);
  float spenning = raaverdi * (5.0 / 1023.0);
  float temperatur = (spenning - 0.5) * 100.0; // TMP36-formel

  Serial.print("Temp: ");
  Serial.print(temperatur);
  Serial.println(" C");

  lcd.setCursor(0, 1);
  lcd.print(temperatur, 1); // 1 desimal
  lcd.print(" C ");

  delay(1000);
}

```

Oppgaver

1. Installer biblioteket LiquidCrystal_I2C via Verktøy → Administrer biblioteker.
2. Koble TMP36 og LCD som vist og last opp koden.
3. Åpne seriell monitor (9600 baud). Ser du temperaturen?
4. Hold fingrene rundt TMP36 – stiger temperaturen?
5. Legg til en alarm: rød LED lyser når temp > 30°C.
6. Utfordring: vis både Celsius og Fahrenheit ($F = C \times 9/5 + 32$).

Løsningsforslag - alarm-LED og Fahrenheit

```

#include <Wire.h>
#include <LiquidCrystal_I2C.h>

LiquidCrystal_I2C lcd(0x27, 16, 2);
const int LED_ALARM = 7;
const float GRENSE = 30.0;

void setup() {
  Serial.begin(9600);
  pinMode(LED_ALARM, OUTPUT);
  lcd.init(); lcd.backlight();
}

void loop() {
  int raa = analogRead(A0);
  float volt = raa * (5.0 / 1023.0);
  float celsius = (volt - 0.5) * 100.0;
  float fahrenheit = celsius * 9.0 / 5.0 + 32.0;

  lcd.setCursor(0, 0);

```

```
lcd.print(celsius, 1); lcd.print((char)223); lcd.print("C ");  
lcd.setCursor(0, 1);  
lcd.print(fahrenheit, 1); lcd.print((char)223); lcd.print("F ");  
  
digitalWrite(LED_ALARM, celsius > GRENSE ? HIGH : LOW);  
delay(1000);  
}
```

Finner ikke LCD-en? Kjør en I2C-scanner sketch for å finne adressen. Adressen er som regel 0x27 eller 0x3F.